The 'Shoot' routine at line 8310 chooses a random point at the bottom of the screen and directs sprite 1 down to it, the process repeating until the player presses a key. The screen colours are reset to normal, the screen is cleared and the sprites are turned off before returning to the main program. To use this subroutine with Digitaya, the following line should be inserted:

```
3845 GOSUB 8000:REM JOYSTICK PORT PICTURE
```

The other listing provides a graphics display for the ALU location in Digitaya and demonstrates different methods of displaying characters on the screen. Lines 7040 to 7090 read a number of DATA statements and POKE the values straight into the screen area. The corresponding location in the colour area is also POKEd with the colour code for the character. In this example the colour code is 2, causing characters to be displayed in red.

A rather unusual trick is used to cause the large letters ALU to scroll down the screen. The first line of graphics character codes that go to make up the letters ALU are POKEd to the second line on the screen. The subroutine at line 7680 is then called, causing the screen to scroll down one line. The second line of codes can then be POKEd into the same screen area as the first, and the subroutine is called again. Repeating this for each of the eight lines of code makes the letters ALU appear to scroll from the top of the screen.

Two other methods of presenting character data to the screen are demonstrated. Characters can be PRINTed directly, as at lines 7130 and 7140, or read as a data string to be PRINTed, as is the case with the question mark design at lines 7170 and 7590 to 7670. This second method allows ease of design within the DATA statements.

To use this routine add the following line:

```
4565 GOSUB 7000: ALU PICTURE
```

Letter Writing
The ALU location for Digitaya is created from three PET low-resolution graphics characters, as shown. The large letters formed appear to scroll down from the top of the screen into their resting position



PET Graphics Character	Screen Codes Normal	Reverse
□	32	160
▣	105	233
▫	95	223

LIZ DIXON

ALU Screen

```
7000 REM **** ALU PICTURE S/R ****
7010 VIC=36864:SC=7680:CS=55296:CC=1024
7020 PRINT CHR$(147):REM CLEAR SCREEN
7030 POKE VIC+32,0:POKE VIC+33,0:REM SET SCREEN/BORD
7040 CC=0
7050 FOR J=1 TO 8:GOSUB7680:REM SCROLL
7060 FOR I=47 TO 72
7070 READ A:CC=CC+A:POKE SC+I,A:POKE CS+I,2
7080 NEXT I,J
7090 READ CS:IF CS<>CC THEN PRINT"CHECKSUM ERROR":STOP
7100 GOSUB7680:REM SCROLL
7110 PRINTCHR$(158):REM TEXT YELLOW
7120 FOR I=1 TO 8:PRINT:NEXT I:REM MOVE DOWN
7130 PRINTTAB(9)"AND":SPC(7);"OR":SPC(8);"NOT"
7140 PRINTTAB(10)"0";SPC(9);"0";SPC(9);"0"
7150 PRINTCHR$(28):REM TEST RED
7160 REM ** QUESTION MARK **
7170 FOR I=1 TO 9:READ D$:PRINTTAB(16)D$:NEXT I
7180 REM **** AWAIT KEY AND RESET ****
7190 GET A$:IF A$="" THEN 7190
7200 POKE VIC+32,14:POKE VIC+33,6:REM SCREEN/BORD
7210 PRINTCHR$(154):REM LT BLUE TEXT
7220 PRINTCHR$(147):REM CLEAR SCREEN
7230 RETURN
7240 REM **** SCREEN DATA ****
7250 REM ** ROW 1 **
7260 DATA 160,32,32,32,32,160,32,32,32,32
7270 DATA 95,160,160,160,160,105
7280 DATA 32,32,32,32,95,160,160,160,160,105
7290 REM **ROW 2 **
7300 DATA 160,32,32,32,32,160,32,32,32,32
7310 DATA 160,223,32,32,32,32,32,32,32,32
7320 DATA 160,223,32,32,253,160
7330 REM ** ROW 3 **
7340 DATA 160,32,32,32,32,160,32,32,32,32
7350 DATA 160,32,32,32,32,32,32,32,32,32
7360 DATA 160,32,32,32,32,160
7370 REM ** ROW 4 **
7380 DATA 160,160,160,160,160,160,32,32,32,32
7390 DATA 160,32,32,32,32,32,32,32,32,32
7400 DATA 160,32,32,32,32,160
7410 REM ** ROW 5 **
7420 DATA 160,32,32,32,32,160,32,32,32,32
7430 DATA 160,32,32,32,32,32,32,32,32,32
7440 DATA 160,32,32,32,32,160
7450 REM ** ROW 6 **
7460 DATA 233,105,32,32,95,223,32,32,32,32
7470 DATA 160,32,32,32,32,32,32,32,32,32
7480 DATA 160,32,32,32,32,160
7490 REM ** ROW 7 **
7500 DATA 32,233,105,95,223,32,32,32,32,32
7510 DATA 160,32,32,32,32,32,32,32,32,32
7520 DATA 160,32,32,32,32,160
7530 REM ** ROW 8 **
7540 DATA 32,32,233,223,32,32,32,32,32,32
7550 DATA 160,32,32,32,32,32,32,32,32,32
7560 DATA 160,32,32,32,32,160
7570 DATA 14463:REM CHECKSUM
7580 REM ** QUESTION MARK DATA **
7590 DATA "  ???  "
7600 DATA "??   ??"
7610 DATA "??   ??"
7620 DATA "    ??  "
7630 DATA "   ??   "
7640 DATA "   ??   "
7650 DATA "   ??   "
7660 DATA "   ??   "
7670 DATA "   **   "
7680 REM **** SCROLL SCREEN S/R ****
7690 POKE 218,160
7700 PRINTCHR$(19);CHR$(17);CHR$(157);CHR$(148)
7710 RETURN
```

Joystick Port Screen

```
8000 REM **** JOYSTICK PICTURE S/R ****
8010 PRINTCHR$(147):REM CLEAR SCREEN
8020 S0=832:S1=S0+64:REM SPR DATA START ADDR
8030 CC=0:VIC=53248:REM START OF VIC CHIP
8040 FOR I=S0 TO S0+62:READ A:CC=CC+A:POKE I,A:NEXT
8050 FOR I=S1 TO S1+62:POKE I,0:NEXT I
8060 FOR I=S1+25 TO S1+27:READ A:CC=CC+A:POKE I,A:NEXT I
8065 READ CS:IF CS<>CC THEN PRINT"CHECKSUM ERROR":STOP
8070 POKEVIC+33,0:POKEVIC+32,0:REM SET SCREEN/BORDER
8080 REM ** SET SPRITE POINTERS **
8090 POKE 2040,S0/64:POKE 2041,S1/64
8100 REM ** SET VIC SPRITE CONTROL REGS **
8110 POKE VIC+39,7:REM SET SPR 0 COLOUR
8120 POKE VIC+40,7:REM SET SPR 1 COLOUR
8130 POKE VIC,65:REM SET SPR 0 X COORD
8140 POKE VIC+1,70:REM SET SPR 0 Y COORD
8150 POKE VIC+2,74:REM SET SPR 1 X COORD
8160 POKE VIC+3,76:REM SET SPR 1 Y COORD
8170 POKE VIC+29,1:REM EXPAND SPR 0 HORIZ
8180 LE$="":FOR I=1 TO 40:LE$=LE$+CHR$(195):NEXT I
8200 DW$="":FOR I=1 TO 25:DW$=DW$+CHR$(17):NEXT I
8210 LS$="":FOR I=1 TO 11:LS$=LS$+CHR$(206)+" ":NEXT I
8220 RS$="":FOR I=1 TO 11:RS$=RS$+CHR$(205)+" ":NEXT I
8230 PRINTCHR$(158):REM TEXT YELLOW
8240 PRINTCHR$(19):PRINT TAB(2)"JOYSTICK PORT"
8250 PRINTCHR$(154):REM TEXT LT BLUE
8260 PRINTCHR$(19)LEFT$(DW$,17)LE$
8270 PRINT CHR$(145);
8280 PT$=LEFT$(LS$,10)+LEFT$(RS$,21)
8290 PT$=PT$+RIGHT$(LS$,10)+RIGHT$(RS$,21)
8300 FOR I=1 TO 3:PRINT PT$;:NEXT I
8305 POKE VIC+21,3:REM TURN ON SPR 0 & 1
8310 REM ** SHOOT **
8320 YB=240:Y1=70
8330 X1=74:N=INT(RND(1)*150)+24
8340 G=2*(X-X1)/(YB-Y1)
8350 FOR Y=Y1 TO YB STEP 2
8360 X1=X1+G:POKE VIC+2,X1:POKE VIC+3,Y
8370 NEXT Y
8380 GET A$:IF A$="" THEN 8330
8390 REM ** RESET SCREEN **
8400 POKE VIC+21,0:REM TURN SPRITES OFF
8410 POKE VIC+32,14:POKEVIC+33,6:REM RESET SCREEN/BORD
8420 PRINT CHR$(147):REM CLEAR SCREEN
8430 RETURN
8440 REM **** SPRITE DATA ****
8450 DATA 0,0,0,63,255,252,64,0,2,64,0,2
8460 DATA 128,0,1,128,0,1,162,16,133,162,16,133
8470 DATA 128,0,1,128,0,1,128,0,1,169,0,1
8480 DATA 128,0,1,128,0,1,136,66,17,72,66,18
8490 DATA 64,0,2,64,0,2,32,0,4,31,255,248
8495 DATA 0,0,0,16,0,0,56,0,0,124,0,0,56,0,0,16
8497 DATA 3701:REM CHECKSUM
```